
Scribe: Elliot Anshelevich Lecturer: Andrew Myers

1 Definition of Lambda Calculus

So far we have only looked at IMP, which has no functions. Now, we will look at another language, known
as Lambda Calculus, which is all functions. Here is a context-free grammar of this language:

e ::= x | e0 e1 | λ x e0

where

x is an Identifier. This refers to a variable defined by surrounding context.

e0 e1 is an Application. Here, e0 is a function and e1 is the argument given to it, so e0 e1 applies the
function e0 to the argument e1.

λ x e0 is an Abstraction/lambda term. This defines a new function with argument variable x and body
e0 (something like ML’s fn x ⇒ e0)

The Lambda Calculus is actually a notation for writing down mathematical functions, but we can also treat
it as a universal, simple core language. Note that while Lisp and Scheme are based somewhat on Lambda
Calculus, there are differences as well.

So, what is a valid program in Lambda Calculus? To answer that, we must first define open and closed
terms. A term is an expression denoting a value. For example, something like int in C or Java would not be
a term. A closed term is a term where all identifiers are bound by the closest containing abstraction. For
example, in the term (λ x ... x (λ y ... y ... ) ... ), the y is bound to the inner lambda term and x to the
outer. An open term is a term that is not closed, i.e. where there are some identifiers that are not bound
to anything. For example, the term (λ x (y x)) is open, since y is not bound to anything in this term. Now
we can finally define the set of legal Lambda Calculus programs. This is just the set of all closed terms.

To fully define Lambda Calculus, we must still define the evaluation rules for it. In Lambda Calculus,
we consider functions to be values, which means that a lambda term evaluates to itself, since it is already a
final value. Therefore, we only need to define evaluation rules for applications. Applications are evaluated
by a rule known as β-reduction:

((λ x e1) e2) → e1{e2/x}
where e1{e2/x} means “e1 with e2 substituted for occurrences of x”. Note that defining “substituted” can
be rather tricky. Here are some examples of β-reduction:

((λ x x) e) →x{e/x}=e
((λ x (λ x x)) e) →(λ x x){e/x} = (λ x x)

(((λ x (λ y (y x ))) 3) INC) →((λ y (y 3)) INC) →(INC 3) →4

In the above examples, we have used INC (the functions which takes an integer as an argument and adds
1 to it) and 3. However, we have no numbers appearing in our grammar for Lambda Calculus. In fact, we
shall see how to form numbers from closed lambda terms later.

2 Functions With Multiple Arguments

〉From the constructs we defined in the previous section, we can form many others. We shall start with
higher-order functions. With Lambda Calculus, we can express functions which return or accept other
functions easily (since all values are only functions). For example, here is a function which applies another
function to 5 and returns the result: (λ f (f 5)). And here is a function that returns a function that applies
another function to its argument: (λ v (λ f (f v))). Applied to 5, it gives us the previous function.

What about functions which take multiple arguments? The grammar we defined for Lambda Calculus
only allows for functions which take one argument. To allow for multiple arguments, we would apparently
need something like:

e ::= ... | e0 e1...en | λ (x1...xn) e0

Here, λ (x1...xn) e0 is a function which takes n arguments x1...xn, and has the body e0. e0 e1...en denotes
an application of a function e0 which takes n − 1 arguments to the arguments e1...en. However, we do not
need these additions and can treat multiple-argument application and abstraction as convenient syntactic
sugar. We can desugar (trivially rewrite syntactically) these terms into the single-argument calculus:

(λ (x1...xn)e) ⇒ (λ x1(λ ... (λ xn e)...))
(e0 e1 e2 ... en) ⇒ (...((e0 e1) e2)... en)

In this way, multi-argument functions are curried (applied one argument at a time):

(+ 1 5) ⇒ ((+ 1) 5)

Notice that (+ 1 5) is really just a shorthand (syntactic sugar) for ((+ 1) 5). Here is another example:

(((λ x (λ y (y x ))) 3) INC) →((λ y (y 3)) INC) →(INC 3) →4
Shorthand: (λ (x y) ( y x )) = (λ x (λ y ( y x )))

((λ (x y) ( y x )) 3 INC) →(INC 3) →4

3 Operational Semantics

Now we shall consider an operational semantics for Lambda Calculus. The configuration is just an expression
of the language, since we have no store (the state is determined entirely by the expression). In large-step
semantics, we have the following inference rule:

e0 ⇓ λ x e2 e2{e1/x} ⇓ v

e0 e1 ⇓ v

This rule can actually be applied in several different ways. Using call-by-name semantics, we have that
e1 is not evaluated before substitution. However, we could also use call-by-value semantics, in which case
arguments are fully evaluated before substituting them into the body of the function. In either case, we
have that any lambda term is a value: v ::= λ x e. Therefore, if we wanted to write an evaluation rule for
a lambda term, it would just evaluate to itself, since it is a value. The only other possible case other than
a lambda term or an application is a single identifier. However, that is not a valid program since it is not
closed.

As for small-step semantics, we have the following rules for call-by-name:

(λ x e1) e2 → e1{e2/x} (β-reduction)

e1 → e′1
e1 e2 → e′1 e2

To model call-by-value semantics, we instead have the following rules:

(λ x e1) v → e1{v/x} (β-reduction)

e2 → e′2
v e2 → v e′2

e1 → e′1
e1 e2 → e′1 e2

These rules require that an expression must be fully evaluated before it can be substituted in a β reduction.

4 Some More Constructs

Since Lambda Calculus is Turing complete, there must be a way to write an infinite loop in it. Here is an
infinite loop:

LOOP � (λ x (x x))(λ x (x x)) → ?

This expression diverges (never stops taking small steps), since, as the reader can easily check, this expression
evaluates to itself: LOOP→LOOP.

When looking at a language like this, you might start missing some constructs that you are used to.
However, some of them are not really needed, since they can be simulated using our current constructs. For
example, Lambda Calculus has no “let” statement like ML does. However, we can simulate (desugar) a let
statement in the following way:

(let x = e1 in e2) =⇒ (λ x e2) e1

Lambda calculus terms can become long. For compactness we will use certain names, as well as multiple
arguments, as shorthand. These are not actually part of the language. Here are some definitions for names
we will use:

IDENTITY � (λ x x)
INC � (+ 1)

APPLY-TO-FIVE � (λ f(f 5))
COMPOSE � (λ (f g)(λ x (f (g x))))

TWICE � (λ f (COMPOSE f f))

Here, COMPOSE composes two functions, and TWICE returns a function that calls the given function twice.
For example:

((COMPOSE INC INC) 2)→ 4
((TWICE (TWICE INC)) 0)→ 3

Lambda Calculus is universal. This means that no primitive boolean type or “if” statement is needed. We
can form them as follows:

TRUE � (λ x (λ y x)) ∼ (λ (x y) x)
FALSE � (λ x (λ y y)) ∼ (λ (x y) y)
if e1 then e2 else e3 ⇒ (IF e1 e2 e3)

IF � (λ (x y z)(x y z))

So, TRUE is a function which takes two arguments and returns the first one, and FALSE returns the second
one. Here is why IF works:

(IF TRUE e2 e3) → (((λ x (λ y x)) e2) e3) → ((λ y e2) e3) → e2

IF works similarly if the first argument to it evaluates to FALSE. Note that call-by-name here is important!
e2 and e3 are not evaluated eagerly by IF.

We can also represent pairs and lists. The pair/list operations are:

(CONS x y) : construct a list with head x and tail y

(FIRST p) : return first item in list (or first item in pair)

(REST p) : return remainder of list (or second item in pair)

Here is one way to implement these operations:

CONS � (λ (x y)(λ f (f x y)))
FIRST � (λ p (p (λ (x y) x))) = (λ p (p TRUE))
REST � (λ p (p (λ (x y) y))) = (λ p (p FALSE))

Another structure which we definitely need is the natural numbers. We can model the number n as a function
that composes an arbitrary function n times. These numbers are called Church numerals. Here is what they
look like:

0 � (λ (f a) a) (=FALSE)

1 � (λ (f a) (f a))

2 � (λ (f a) (f (f a)))

3 � (λ (f a) (f (f (f a))))

n � (λ (f a) (f (...(f a)...)))

We can now define the INC function, that adds one to a number, by writing a function that interposes an
extra call to the function as follows:

n � (λ (f a)(fn a)) , so
nf = (λ a (fn a)) , and

f((n f) a) = fn+1 a , therefore
INC � (λ n (λ (f a)(f((n f) a))))

We can now define + and other arithmetic operators, by using the same trick:

+ � (λ (n1 n2) (λ (f a)((n1 f)((n2 f) a)))) or
+ � (λ (n1 n2) ((n1 INC) n2))

